--# -path=alltenses
-- -path=.:lib-3.1.6/prelude::lib-3.1.6/abstract:lib-3.1.6/common:lib-3.1.6/scandinavian:lib-3.1.6/swedish:lib-3.1.6/api

concrete Bamse of BamseAbs = Svenska ** 
  open AllSwe, ParadigmsSwe, IrregSwe, Prelude, (M=MorphoSwe), ParamX in {

flags coding = utf8;

lin 

den_det = DetQuant DefArt NumSg;
en_ett = DetQuant IndefArt NumSg;
de = DetQuant DefArt NumPl;
pl_indef = DetQuant IndefArt NumPl;
nagon_t = someSg_Det;
nagra = somePl_Det;

vilken_t = IdetQuant which_IQuant NumSg;
vilka = IdetQuant which_IQuant NumPl;

som = IdRP;

liten = small_A;
lustig = mkA "lustig";
stark = mkA "stark";

pa = mkPrep "på";
i = mkPrep "i";
med = mkPrep "med";
av = mkPrep "av";

han = he_Pron;
hon = she_Pron;
jag = i_Pron;
du = youSg_Pron;
dom = they_Pron;
den = M.regNP "den" "dens" M.Neutr Sg;
det = it_Pron;

kulle = mkN "kulle" utrum;
sol = sun_N;
hus = house_N;
van = friend_N;
burk = mkN "burk" utrum;
honung = mkN "honung" utrum;
dag = mkN "dag" utrum;
dunderhonung = mkN "dunderhonung" utrum;
varld = mkN "värld" utrum;
bjorn = mkN "björn" utrum;

dar = there_Adv;

alltid = always_AdV;
aldrig = ss "aldrig";
inte = ss "inte";

ligger_still = mkV (mkV "ligger") "still";
ligger = mkV2 ligga_V;
bor = mkV2 (mkV "bor");
tittar = mkV2 (mkV "tittar");
borjar = mkV2 (mkV "börjar");
blir = mkV2 bliva_V;
ater = mkV "äta" "åt" "ätit";
ar = mkV2 M.verbBe;

och = and_Conj;
men = {s1 = []; s2 = ["men"] ;n = Sg};
komma_och = {s1 = []; s2 = [", och"]; n = Pl};
komma_men = {s1 = []; s2 = [", men"]; n = Sg};

nar = when_Subj;

bamse = mkPN "Bamse";
skalman = mkPN "Skalman";
lille_skutt = mkPN ["Lille Skutt"];
farmor = mkPN "Farmor";

}
